Здравствуйте. Есть запрос, в котором суммирую количество часов и сумму подзаданий к их родительской задаче. Но если подзаданий больше чем одно, то считает сумму не вверно(Каждое подзадание суммирует с основным заданием)
Как правильно построить запрос?

```1c
    ВЫБРАТЬ
    СУММА(ЗаказПокупателяРасшифровка.КоличествоЧасов + ЕСТЬNULL(ЗаказПокупателяРасшифровка1.КоличествоЧасов, 0)) КАК КоличествоЧасов,
    ЗаказПокупателяРасшифровка.Задание,
    ЗаказПокупателяРасшифровка.Исполнитель,
    СУММА(ЗаказПокупателяРасшифровка.Цена + ЕСТЬNULL(ЗаказПокупателяРасшифровка1.Цена, 0)) КАК Цена
    ИЗ
    Документ.ЗаказПокупателя.Расшифровка КАК ЗаказПокупателяРасшифровка
    ЛЕВОЕ СОЕДИНЕНИЕ Документ.ЗаказПокупателя.Расшифровка КАК ЗаказПокупателяРасшифровка1
    ПО ЗаказПокупателяРасшифровка.ID = ЗаказПокупателяРасшифровка1.IDРодителя
    ГДЕ
    ЗаказПокупателяРасшифровка.Ссылка = &ТекущийДокумент
    И ЗаказПокупателяРасшифровка.IDРодителя = 0

    СГРУППИРОВАТЬ ПО
    ЗаказПокупателяРасшифровка.Исполнитель,
    ЗаказПокупателяРасшифровка.Задание
```
---
Необходимо использовать не левое соединение, а объединение. Примерно так:

```1c
    ВЫБРАТЬ
    ВложенныйЗапрос.Задание КАК Задание,
    ВложенныйЗапрос.Исполнитель КАК Исполнитель,
    СУММА(ВложенныйЗапрос.КоличествоЧасов) КАК КоличествоЧасов,
    СУММА(ВложенныйЗапрос.Цена) КАК Цена
    ИЗ
    (ВЫБРАТЬ
    РасшифровкаЗадач.Задание КАК Задание,
    РасшифровкаЗадач.Исполнитель КАК Исполнитель,
    РасшифровкаЗадач.КоличествоЧасов КАК КоличествоЧасов,
    РасшифровкаЗадач.Цена КАК Цена
    ИЗ
    Документ.ЗаказПокупателя.Расшифровка КАК РасшифровкаЗадач
    ГДЕ
    РасшифровкаЗадач.Ссылка = &ТекущийДокумент
    И РасшифровкаЗадач.IDРодителя = 0

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    РасшифровкаЗадач.Задание КАК Задание,
    РасшифровкаЗадач.Исполнитель КАК Исполнитель,
    РасшифровкаПодзадач.КоличествоЧасов КАК КоличествоЧасов,
    РасшифровкаПодзадач.Цена КАК Цена
    ИЗ
    Документ.ЗаказПокупателя.Расшифровка КАК РасшифровкаЗадач
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ Документ.ЗаказПокупателя.Расшифровка КАК РасшифровкаПодзадач
    ПО РасшифровкаЗадач.Ссылка = РасшифровкаПодзадач.Ссылка
    И РасшифровкаЗадач.ID = РасшифровкаПодзадач.IDРодителя
    ГДЕ
    РасшифровкаЗадач.Ссылка = &ТекущийДокумент
    И РасшифровкаЗадач.IDРодителя = 0
    )
    КАК ВложенныйЗапрос

    СГРУППИРОВАТЬ ПО
    ВложенныйЗапрос.Задание,
    ВложенныйЗапрос.Исполнитель
```
---
СУММА(ЗаказПокупателяРасшифровка.КоличествоЧасов + ЕСТЬNULL(ЗаказПокупателяРасшифровка1.КоличествоЧасов, 0)) - можете объяснить что тут вычисляется?
---
Это я так подзадачу с задачей суммировал..
---
Необходимо использовать не левое соединение, а объединение. Примерно так:

```1c
    ВЫБРАТЬ
    ВложенныйЗапрос.Задание КАК Задание,
    ВложенныйЗапрос.Исполнитель КАК Исполнитель,
    СУММА(ВложенныйЗапрос.КоличествоЧасов) КАК КоличествоЧасов,
    СУММА(ВложенныйЗапрос.Цена) КАК Цена
    ИЗ
    (ВЫБРАТЬ
    РасшифровкаЗадач.Задание КАК Задание,
    РасшифровкаЗадач.Исполнитель КАК Исполнитель,
    РасшифровкаЗадач.КоличествоЧасов КАК КоличествоЧасов,
    РасшифровкаЗадач.Цена КАК Цена
    ИЗ
    Документ.ЗаказПокупателя.Расшифровка КАК РасшифровкаЗадач
    ГДЕ
    РасшифровкаЗадач.Ссылка = &ТекущийДокумент
    И РасшифровкаЗадач.IDРодителя = 0

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    РасшифровкаЗадач.Задание КАК Задание,
    РасшифровкаЗадач.Исполнитель КАК Исполнитель,
    РасшифровкаПодзадач.КоличествоЧасов КАК КоличествоЧасов,
    РасшифровкаПодзадач.Цена КАК Цена
    ИЗ
    Документ.ЗаказПокупателя.Расшифровка КАК РасшифровкаЗадач
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ Документ.ЗаказПокупателя.Расшифровка КАК РасшифровкаПодзадач
    ПО РасшифровкаЗадач.Ссылка = РасшифровкаПодзадач.Ссылка
    И РасшифровкаЗадач.ID = РасшифровкаПодзадач.IDРодителя
    ГДЕ
    РасшифровкаЗадач.Ссылка = &ТекущийДокумент
    И РасшифровкаЗадач.IDРодителя = 0
    )
    КАК ВложенныйЗапрос

    СГРУППИРОВАТЬ ПО
    ВложенныйЗапрос.Задание,
    ВложенныйЗапрос.Исполнитель
```
---
спасибо, кажется, верно